
Review Goals

Refresh large-n inference for means: test statistic, p-value, confidence intervals (normal
approximation)

Revisit the Simple Linear Regression Model (SLRM): assumptions, OLS estimator, sampling
variability

Revisit Multiple Linear Regression Models (MLRM): ceteris paribus interpretation and omitted
variable bias intuition

Practice OLS inference: (asymptotic) t/z tests, confidence intervals, and goodness-of-fit (R2, R̄2)

Joint hypotheses / exclusion restrictions: restricted vs unrestricted models and the F -statistic

Translate each step into (i) built-in Stata commands and (ii) “by hand” computations

1 / 34



Foundations: Random Variables and Distributions
Population Parameter (unknown):

µY = E (Y ) =

∫
y · f (y) dy (population mean)

Population Variance (measures spread):

σ2
Y = Var(Y ) = E [(Y − µY )

2]

Covariance (linear relationship):

Cov(X ,Y ) = E [(X − µX )(Y − µY )]

Correlation (scaled covariance):

Corr(X ,Y ) =
Cov(X ,Y )

σXσY
, −1 ≤ Corr(X ,Y ) ≤ 1

Random Sampling: Y1,Y2, . . . ,Yn are i.i.d. (independently and identically distributed)

Sample Mean: Estimation and Properties (1)

Sample Mean (unbiased estimator of µY ):

Y =
1

n

n∑
i=1

Yi

Key Properties:

Unbiasedness: E (Y ) = µY

Variance of Y : Var(Y ) =
σ2
Y

n

Standard Error (estimated): SE(Y ) = sY√
n

where

s2Y =
1

n − 1

n∑
i=1

(Yi − Y )2

is the sample variance

Sample Mean: Estimation and Properties (2)

Large-n inference (CLT / normal approximation):

t ≡ Y − µ0

sY /
√
n

a−→
H0

N(0, 1)

95% CI (large n):

Y ± 1.96 · sY√
n

Sample Mean Inference in Stata (1)

Estimate mean and standard error:

// Load data

use firm_data.dta , clear

// Descriptive statistics

summarize sales_employee fwp

// Standard error of mean (manual)

display "SE(mean) = " %6.4f r(sd)/sqrt(r(N))

Hypothesis test (Stata built-in): H0 : µY = µ0

ttest sales_employee = 230

Sample Mean Inference in Stata (2)

Confidence interval (manual large-n):

summarize sales_employee

scalar ybar = r(mean)

scalar sd = r(sd)

scalar n = r(N)

display "95% CI: [" %6.2f (ybar - 1.96*sd/sqrt(n)) ///

", " %6.2f (ybar + 1.96* sd/sqrt(n)) "]"

Manual: Difference-in-Means (Two Groups) (1)

Statistic:

t =
Y 1 − Y 2√

s21/n1 + s22/n2
(large n: t ≈ N(0, 1))

// Numerical example

scalar n1 = 238

scalar n2 = 238

scalar mean1 = 238.49

scalar mean2 = 223.96

scalar sd1 = 149.09

scalar sd2 = 101.63

Manual: Difference-in-Means (Two Groups) (2)

scalar se_diff = sqrt((sd1 ^2)/n1 + (sd2 ^2)/n2)

scalar t_stat = (mean1 - mean2) / se_diff

// Large -n two -sided p-value

scalar pval = 2 * (1 - normal(abs(t_stat)))

display "t = " %6.3f t_stat

display "p-value = " %6.4f pval

8 / 34



Manual: Confidence Interval for Difference in Means

95% CI (large n):

(Y 1 − Y 2)± 1.96 · SE (Y 1 − Y 2)

scalar diff = mean1 - mean2

scalar ci_l = diff - 1.96* se_diff

scalar ci_u = diff + 1.96* se_diff

display "Diff = " %6.2f diff

display "95% CI: [" %6.2f ci_l ", " %6.2f ci_u "]"

Key rule: CI and two-sided tests are equivalent.

Simple Linear Regression Model (SLRM)

Population Regression Function:

Yi = β0 + β1Xi + εi

Key assumption (mean independence):

E (εi | Xi ) = 0 ⇒ E (Yi | Xi ) = β0 + β1Xi

Other common assumptions:

Random sampling: (Yi ,Xi ) are i.i.d.

No perfect collinearity: X varies in the sample

Homoskedasticity (when used): Var(εi | Xi ) = σ2

Interpretation: β1 is the change in E (Y | X ) for a one-unit increase in X .

OLS Estimators

OLS minimizes the Sum of Squared Residuals (SSR):

min
β0,β1

n∑
i=1

(Yi − β0 − β1Xi )
2

OLS slope and intercept:

β̂1 =

∑n
i=1(Xi − X )(Yi − Y )∑n

i=1(Xi − X )2
, β̂0 = Y − β̂1X

Key properties (under standard assumptions):

Unbiasedness: E (β̂1) = β1

Consistency: β̂1
p−→ β1

Simple Regression: Stata

Estimate SLRM:

regress sales_employee fwp

Test on slope:

test fwp = 0

// Access stored results

display "b = " _b[fwp]

display "se = " _se[fwp]

display "z/t = " _b[fwp]/_se[fwp]

Fitted values and residuals:

predict yhat , xb

predict ehat , resid

summarize yhat ehat

Manual Computation: OLS Slope from Summary Stats (1)

Using correlation identity (simple regression):

β̂1 =
Cov(X ,Y )

Var(X )
= ρXY · sY

sX

// Get sd(y), sd(x), corr(x,y)

summarize sales_employee

scalar sd_y = r(sd)

summarize fwp

scalar sd_x = r(sd)

correlate sales_employee fwp

matrix C = r(C)

scalar corr_xy = C[1,2]

Manual Computation: OLS Slope from Summary Stats (2)

// beta1 = corr * sd_y / sd_x

scalar beta1 = corr_xy * (sd_y/sd_x)

// intercept: beta0 = ybar - beta1*xbar

summarize sales_employee , meanonly

scalar ybar = r(mean)

summarize fwp , meanonly

scalar xbar = r(mean)

scalar beta0 = ybar - beta1*xbar

display "beta1 (manual) = " %8.4f beta1

display "beta0 (manual) = " %8.4f beta0

Manual Computation: σ̂2 and SE(β̂1) (SLRM) (1)

// After: regress y x

// (here y = sales_employee , x = fwp)

predict yhat , xb

gen ehat = sales_employee - yhat

// SSR = sum ehat^2

gen ehat2 = ehat^2

quietly summarize ehat2 , meanonly

scalar SSR = r(sum)

// n and k (SLRM has k=2: constant + x)

scalar n = e(N)

scalar k = 2

Manual Computation: σ̂2 and SE(β̂1) (SLRM) (2)

// sigma^2 hat = SSR/(n-k) = SSR/(n-2)

scalar sig2hat = SSR/(n-k)

// SSX = sum (x - xbar)^2

quietly summarize fwp , meanonly

scalar xbar = r(mean)

gen xdev2 = (fwp - xbar)^2

quietly summarize xdev2 , meanonly

scalar SSX = r(sum)

// Var(b1) = sigma ^2 / SSX ; SE = sqrt(Var)

scalar var_b1 = sig2hat/SSX

scalar se_b1 = sqrt(var_b1)

display "SSR = " SSR

display "sig2hat = " sig2hat

display "Manual SE(b1) = " se_b1

Manual: Test and Confidence Interval on Slope (Large n) (1)

Test statistic (usual t-stat; large n normal approximation):

t =
β̂1 − β1,0

SE (β̂1)
≈ N(0, 1)

// After regress

scalar b1 = _b[fwp]

scalar se1 = _se[fwp]

scalar b10 = 0

scalar tstat = (b1 - b10)/se1

scalar pval = 2*(1 - normal(abs(tstat)))

Manual: Test and Confidence Interval on Slope (Large n) (2)

display "t/z = " %6.3f tstat

display "p-value = " %6.4f pval

// 95% CI (large -n)

scalar ci_l = b1 - 1.96* se1

scalar ci_u = b1 + 1.96* se1

display "95% CI: [" %6.3f ci_l ", " %6.3f ci_u "]"

Goodness of Fit: R2

Decomposition:

TSS = ESS + SSR

TSS =
∑n

i=1(Yi − Y )2

SSR =
∑n

i=1 ε̂
2
i

ESS =
∑n

i=1(Ŷi − Y )2 = TSS− SSR

R2:

R2 =
ESS

TSS
= 1− SSR

TSS

In simple regression: R2 = Corr(X ,Y )2.

Manual Computation: R2 via TSS and SSR (1)

// After: regress y x

predict yhat , xb

gen ehat = sales_employee - yhat

// SSR

gen ehat2 = ehat^2

quietly summarize ehat2 , meanonly

scalar SSR = r(sum)

Manual Computation: R2 via TSS and SSR (2)

// TSS = sum (y - ybar)^2

quietly summarize sales_employee , meanonly

scalar ybar = r(mean)

gen ydev2 = (sales_employee - ybar)^2

quietly summarize ydev2 , meanonly

scalar TSS = r(sum)

// ESS = TSS - SSR

scalar ESS = TSS - SSR

scalar R2 = ESS/TSS

display "R2 (manual) = " R2

Multiple Linear Regression Model (MLRM)

Yi = β0 + β1X1i + · · ·+ βkXki + εi

Key assumption:

E (εi | X1i , . . . ,Xki ) = 0

Interpretation: βj is the ceteris paribus effect of Xj on E (Y | X ).

Omitted Variable Bias (intuition, 2 regressors):

Bias(β̂1) = β2 ·
Cov(X1,X2)

Var(X1)

Multiple Regression: Stata (1)

Estimate MLRM:

regress sales_employee fwp mgmt us_location

Single coefficient test:

test fwp = 0

Joint hypothesis test:

test mgmt us_location

Multiple Regression: Stata (2)

LR test for OLS:

// Store models

regress sales_employee fwp

estimates store r

regress sales_employee fwp mgmt us_location

estimates store ur

// LR test compares Gaussian log -likelihoods

lrtest ur r

Note: In linear regression, LR tests correspond to comparing restricted vs unrestricted fits under a
Gaussian likelihood.

F-statistic for Joint Restrictions (1)

Null: q exclusion restrictions.

Restricted vs Unrestricted:

Restricted (under H0): SSR0

Unrestricted (under H1): SSR1

q = number of restrictions

k = number of regressors including the constant

Note: In some textbooks k denotes the number of slope regressors only, and the denominator is
written as SSR1/(n − k − 1). The two formulas are numerically identical; only the notation differs.

F-statistic for Joint Restrictions (2)

Statistic:

F =
(SSR0 − SSR1)/q

SSR1/(n − k)

Asymptotic distribution under H0 (large n):

F
a−→
H0

χ2
q

q

Intuition: how much the SSR increases when moving from the unrestricted to the restricted model,
relative to the unexplained variance in the unrestricted model.

Manual Computation: F from SSR0 and SSR1

// Example: test mgmt=0 and us_location =0 jointly

// Restricted

regress sales_employee fwp

scalar SSR0 = e(rss)

// Unrestricted

regress sales_employee fwp mgmt us_location

scalar SSR1 = e(rss)

scalar n = e(N)

// k includes the constant

scalar k = 1 + 3 // constant + (fwp mgmt us_location)

scalar q = 2 // restrictions: mgmt=0 and us_location =0

scalar Fstat = ((SSR0 - SSR1)/q) / (SSR1/(n - k))

display "F = " %8.4f Fstat

Dummy Variables

Binary regressor: US = 1 if firm is in US, else 0.

regress sales_employee fwp us_location

Interpretation:

For US = 0: E (Y | X ,US = 0) = β0 + β1X

For US = 1: E (Y | X ,US = 1) = (β0 + β2) + β1X

β2 is the ceteris paribus mean difference (US vs non-US)

Multiple categories: omit one category (reference) to avoid the dummy trap.

Stata Command Summary

Task Stata Command

Summary statistics summarize y x1 x2

Correlation correlate y x1 x2

Difference-in-means ttest y, by(group)

Simple regression regress y x

Multiple regression regress y x1 x2 x3

Test single coef test x1 = 0

Joint test (F) test x2 x3

Fitted values predict yhat, xb

Residuals predict ehat, resid

Model comparison (if shown) lrtest ur r

Key Formulas

□ Sample mean: Y = 1
n

∑
Yi , SE (Y ) = sY√

n

□ Mean test statistic: t = Y−µ0

sY /
√
n
≈ N(0, 1) (large n)

□ 95% CI (large n): Y ± 1.96 · SE (Y )

□ OLS slope (SLRM): β̂1 =
∑

(xi−x̄)(yi−ȳ)∑
(xi−x̄)2

□ Error variance (SLRM): σ̂2 = SSR
n−2 (more generally SSR/(n − k))

□ Regression test statistic: t =
β̂j−βj,0

SE(β̂j )
≈ N(0, 1) (large n)

□ R2: R2 = 1− SSR
TSS

□ F-statistic:

F =
(SSR0 − SSR1)/q

SSR1/(n − k)
a−→

χ2
q

q

Common Mistakes & Sanity Checks

Common mistakes:

1. SSR > TSS: impossible (check computations)

2. R2 > 1 or R2 < 0: computational error

3. All dummies included: perfect multicollinearity

4. Correlation ̸= causation

5. Forgetting df: use (n − k) with k incl. constant

6. Mixing finite-sample vs large-n critical values without stating it

Sanity checks:

▶ Do signs make economic sense?

▶ Are magnitudes reasonable?

▶ Are SEs smaller when n is larger (all else equal)?

▶ Does adding controls change β̂1 in the direction predicted by OVB?

31 / 34



Tips

1. Use large-n critical values: 1.96 for 5% two-sided

2. Confidence intervals and two-sided tests are equivalent

3. In MLRM, coefficients are ceteris paribus effects

4. Joint hypotheses: use restricted vs unrestricted (F test)

5. Dummy variables: always define a reference group

6. OVB sign depends on β2 and Cov(X1,X2)

7. R2 is fit, not causality; compare specifications carefully

8. Always check degrees of freedom: n − k with k incl. constant

Worked Example: Complete Analysis (1)

Question: Effect of family-friendly policies on productivity, with controls?

// 1. Explore

summarize sales_employee fwp mgmt us_location

correlate sales_employee fwp mgmt us_location

// 2. Simple regression

regress sales_employee fwp

estimates store r

// 3. Multiple regression

regress sales_employee fwp mgmt us_location

estimates store ur

Worked Example: Complete Analysis (2)

// 4. Joint test (built -in)

test mgmt us_location

// 5. LR test (if shown in notes for OLS)

lrtest ur r

// 6. "By hand" sanity check (t/z)

scalar tstat = _b[fwp]/_se[fwp]

display "t/z = " tstat

